
How to (X+ EXCEL reports) 

In this tutorial we will create an EXCEL template for X+ from scratch. 

Preparation 
1. Load a Mastercam file with multiple operations using multiple tools 

2. Launch the X+ setup sheet  function 

3. For this tutorial setup the dialog header as below 

4. Setup the parameter as below 

 
 

5. Save the settings as a new configuration (EXCEL_Tutorial) 

Do not hit the OK button yet! 

6. Open Windows-Explorer and navigate to: 

C:\Users\Public\Documents\X+\templates\setup sheets\xlt 

7. Delete EXCEL_Tutorial.xlt(m) files if present. 

We want to create an EXCEL-Template from scratch! 

8. Close Windows-Explorer 

9. Now press the OK button in the X+ dialog 

10. Save the EXCEL file as EXCEL_Tutorial on your desktop. 

Save it as XLSM file because we are going to add VBA macros! 

Now make changes to the EXCEL file 
The output is without any formatting. 

This is the raw output to an EXCEL file. 

The screenshot usually hides some information. So let’s resize it and move it aside (Hold shift while resizing). 

 

When creating an EXCEL report, X+ always uses the active sheet for the output. 

If there is no EXCEL template, as in our case, it is always the 1st sheet! 

However, if we use an EXCEL template the active sheet is the sheet that was active when the template 

was saved! Keep that in mind!  

Rename the sheets 
Let’s rename the sheets first. 

1st sheet as dump 

2nd sheet as setupsheet 

3rd sheet as toollist 

 

Save the file (Ctrl + s) 

Create the layout for the setup sheet 
For the layout copy some cells and the screenshot from the dump to the setup sheet. 

Do some formatting to the cells as needed (font, size, alignment, color, etc.), 

Resize the screenshot and insert a logo. 

In this tutorial, I came up with this.

 

 At this point we are ready to set some references! 

For example G3 refers to dump!C1 ,  G4 refers to  dump!C2….etc. Set the references as below. 

 

reference view 
(Ctrl + #) 

Save the file (Ctrl + s). 

Use VBA to copy and paste cells from dump to setupsheet 
Unfortunately the operation list isn’t static. The length of the list depends on the number of operations 

and will probably change with each Mastercam part. Therefore we can’t use references here. 

Instead of using references we will automate this process with the help of VBA. 

Press Alt + F11 to open the Visual Basic editor. Resize the editor to see your EXCEL workbook as well. 

From the menu select Insert then Module. 

 

Then on the right window type in or copy and paste the following code:  

Sub copyOperationList() 
    Dim opListPart1 As Range    'first part for operation list A20:Fn 
    Dim opListPart2 As Range    'second part of operation list O20:Rn 
    Dim opList As Range         'unified list as range 
    Dim prmList As Range        'whole parameter list as range 
    'get the whole parameter list as range 
    Set prmList = Range("dump!A20").CurrentRegion 
    'find first part of operation list 
    Set opListPart1 = Range(prmList.Cells(1, 1), prmList.Cells(prmList.rows.Count, 6)) 
    'find second part of operation list 
    Set opListPart2 = Range(prmList.Cells(1, 15), prmList.Cells(prmList.rows.Count, prmList.Columns.Count)) 
    'build union from part 1 and part 2 
    Set opList = Union(opListPart1, opListPart2) 
    'copy cells to clipboard 
    opList.Copy 
    'paste cells to setupsheet!A20 (only values) 

    Range("setupsheet!A20").PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks _ 
        :=False, Transpose:=False 
    'Add table object 
    Worksheets("setupsheet").ListObjects.Add(xlSrcRange, Range("setupsheet!A20").CurrentRegion, , xlYes).Name = 
_ 
        "OperationList" 
    'set table format style 
    Worksheets("setupsheet").ListObjects("OperationList").TableStyle = "TableStyleLight1" 
End Sub 
 

 

You should come up with this. 

 

Save the file (Ctrl + s). 

Run the macro 
Switch to the EXCEL workbook. 

Select and delete the entire operation list from the setup sheet . 

Back in the Visual Basic editor, click in the code list. The cursor must be inside the macro! 

Then press F5 to run the macro. 

Tip: 

Recording macros is always a good way to find the right EXCEL commands. I do this regularly. 

If everything went well, you should get something like this. 

 

Use VBA to copy, paste and resize the screenshot 
The screenshot should also be automatically copied to the setup sheet, scaled and positioned. 

Copy and paste the following two macros into the Visual Basic editor. 

'helper macro to fit a shape object into a range 
Sub fit(ByRef pic As Shape, ByRef r As Range) 
    Dim PicWtoHRatio As Single 
    Dim CellWtoHRatio As Single 
    PicWtoHRatio = pic.Width / pic.Height 
    CellWtoHRatio = r.Width / r.Height 
    If PicWtoHRatio / CellWtoHRatio > 1 Then 
        pic.Width = r.Width 
        pic.Height = pic.Width / PicWtoHRatio 
    Else 
        pic.Height = r.Height 
        pic.Width = pic.Height * PicWtoHRatio 
    End If 
    pic.Top = r.Top + r.Height / 2 - pic.Height / 2 'vertical center align 
    pic.Left = r.Left + r.Width / 2 - pic.Width / 2 'horizontal center align 
End Sub 
 
'copy the screenshot 
Sub copyScreenshot() 
    Dim screenshot As Shape 'the screenshot 
    Set screenshot = Worksheets("dump").Shapes(1) 'get the screenshot 

    If Not IsNull(screenshot) Then 
        'copy screenshot to clipboard 
        screenshot.Copy 
        'paste to setupsheet 
        Worksheets("setupsheet").Paste Destination:=Worksheets("setupsheet").Range("A1") 
        'get the copied screenshot 
        Set screenshot = Worksheets("setupsheet").Shapes(Worksheets("setupsheet").Shapes.Count) 
        'fit the screenshot to A1:E16 
        fit screenshot, Worksheets("setupsheet").Range("A1:E16") 
    End If 
End Sub 

 

Test the macros 
Switch to the EXCEL workbook. 

Select and delete the screenshot from the setup sheet . 

Back in the Visual Basic editor, click in the code list. The cursor must be inside the macro 

copyScreenshot()! 

Then press F5 to run the macro. 

If everything went well, the screenshot is perfectly copied, scaled and centered to the range A1:E16. 

 

We are done with the setup sheet. 

Now let’s create the layout for the tool list 
For simplicity, copy the header from the setup sheet and change only the title to "Tool List". 

Then copy all cells for the tool list from dump and set the formatting as needed. 

In this tutorial, I came up with this. 

 

 

Create the macros for the tool list 
Copying the tool list via VBA is almost the same as for the setup sheet, so we can copy and modify the 

code a little. However, we also want to sort the tools and delete duplicates. 

Copy and paste the code below to the Visual Basic editor. 

'sort the tools by number 
Sub sortTools() 
    ActiveWorkbook.Worksheets("toollist").ListObjects("ToolList").Sort.SortFields. _ 
        Clear 
    ActiveWorkbook.Worksheets("toollist").ListObjects("ToolList").Sort.SortFields. _ 
        Add Key:=Range("ToolList[[#All],[Tl '#]]"), SortOn:=xlSortOnValues, Order _ 
        :=xlAscending, DataOption:=xlSortNormal 
    With ActiveWorkbook.Worksheets("toollist").ListObjects("ToolList").Sort 
        .Header = xlYes 
        .MatchCase = False 

        .Orientation = xlTopToBottom 
        .SortMethod = xlPinYin 
        .Apply 
    End With 
End Sub 
 
'copy the tool list to toollist 
Sub copyToolList() 
    Dim toolList As Range     'tool list cells 
    Dim prmList As Range      'whole parameter list as range 
    'get the whole parameter list as range 
    Set prmList = Range("dump!A20").CurrentRegion 
    'get the tool list 
    Set toolList = Range(prmList.Cells(1, 5), prmList.Cells(prmList.rows.Count, 14)) 
    toolList.Copy 
    'paste cells to setupsheet!A20 (only values) 
    Range("toollist!A20").PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks _ 
        :=False, Transpose:=False 
         
    'delete duplicate tools 
    Worksheets("toollist").Range("A20").CurrentRegion.RemoveDuplicates Columns:=1, Header:=  xlYes 
         
    'Add table object 
    Worksheets("toollist").ListObjects.Add(xlSrcRange, Range("toollist!A20").CurrentRegion, , xlYes).name = _ 
        "ToolList" 
         
    'set table format style 
    Worksheets("toollist").ListObjects("ToolList").TableStyle = "TableStyleLight1" 
     
    'sort the tools by number 
    sortTools 
End Sub 

 

Time to create a main macro 
To execute all the macros one by one, we create a main macro and make the calls from there. 

We name it format_sheet() , because X+ calls this macro if it exists in the template. So this is the entry 

point! 

Copy and paste this macro to the Visual Basic editor. 

'format sheet 
Sub format_sheet() 
    'create the setup sheet 
    copyOperationList 
    copyScreenshot ActiveWorkbook.Worksheets("setupsheet") 
    'create the tool list 
    copyToolList 
    copyScreenshot ActiveWorkbook.Worksheets("toollist") 
End Sub 

We also need to make some changes to our copyScreenshot() macro. We add a parameter for the target 

sheet. So we can reuse the code for the tool list. 

'copy the screenshot (modified) 
Sub copyScreenshot(ByRef ws As Worksheet) 
    Dim screenshot As Shape 'the screenshot 
    Set screenshot = ActiveWorkbook.Worksheets("dump").Shapes(1) 'get the screenshot 
    If Not IsNull(screenshot) Then 
        'copy screenshot to clipboard 
        screenshot.Copy 
        'paste to ws (setupsheet or toollist) 
        ws.Paste Destination:=ws.Range("A1") 
        'get the copied screenshot 
        Set screenshot = ws.Shapes(ws.Shapes.Count) 
        'fit the screenshot to A1:E16 
        fit screenshot, ws.Range("A1:E16") 
    End If 
End Sub 

 

Run the format_sheet() macro 
Before we execute the format_sheet() macro, we need to clean up our workbook. 

So switch to the workbook and delete the screenshots, operation list and tool list from both sreenshot 

and toollist. 

Switch to the Visual Basic editor and click into the format_sheet macro. The cursor must be inside this 

macro. 

Press F5 to execute the macro. 

Check the result. 

To speed up the execution of the macros we can disallow EXCELs screen-updating during execution. 

We can also add an Error handler, just in case. 

The final format_sheet() macro looks like this. 

'format sheet 
Sub format_sheet() 
    On Error GoTo error 
    'do not allow sreen updating 
    Application.ScreenUpdating = False 
    'create the setup sheet 
    copyOperationList 
    copyScreenshot ActiveWorkbook.Worksheets("setupsheet") 
    'create the tool list 
    copyToolList 
    copyScreenshot ActiveWorkbook.Worksheets("toollist") 
error: 
    'allow sreen updating 
    Application.ScreenUpdating = True 
End Sub 

 

That’s it. 

To use this workbook as an EXCEL template for X+, we need to clean up the workbook and save it as 

XLTM file. 

So switch to the workbook and delete the screenshots, as well as the operation and tool list from both 

the setupsheet and the toollist as before. 



gmccs.de 

 

 

Also delete the whole content from dump (empty sheet here). 

Set dump as active sheet before saving the workbook as XLTM file!

 

Then save it as: 

"C:\Users\Public\Documents\X+\templates\setup sheets\xlt\EXCEL_Tutorial.xltm" 

Final version: download 

https://www.gmccs.de/downloads/xplus/EXCEL_Tutorial.zip

Use the template with X+ 
In the X+ dialog box, load the EXCEL_Tutorial configuration and then press the OK button. 

 

X+ creates a new EXCEL workbook from our EXCEL template. 

After all the information has been transferred to the active sheet, it calls the macro format_sheet(). 

If you make changes to your configuration, you may need to update your EXCEL template as well. Keep 

that in mind! 

Result 

 

 


